

Nonlinear dynamic stochastic general equilibrium models in Stata 16

David Schenck

Senior Econometrician
Stata

2020 Stata Conference
July 31, 2020


Motivation

- Models used in macroeconomics for policy analysis
- Models for multiple time series
- Linking observed variables to latent factors
- Where link is motivated by economic theory
- Methods for bringing theoretical macroeconomic models to the data

Linking data to a model

- We wish to explain inflation and interest rates with a model
- We use a textbook New Keynesian model
- Inflation, interest rates, and (unobserved) output demand are linked to latent state variables
- Simple model, two states: productivity and monetary policy

Data



Model

- Households demand output, given inflation and interest rates:

$$\frac{1}{X_t} = \beta E_t \left[\left(\frac{1}{X_{t+1}} \right) \left(\frac{R_t}{\Pi_{t+1} Z_{t+1}} \right) \right]$$

Model

- Households demand output, given inflation and interest rates:

$$\frac{1}{X_t} = \beta E_t \left[\left(\frac{1}{X_{t+1}} \right) \left(\frac{R_t}{\Pi_{t+1} Z_{t+1}} \right) \right]$$

- Firms set prices, given output demand:

$$\phi + (\Pi_t - 1) = \frac{1}{\phi} X_t + \beta E_t [\Pi_{t+1} - 1]$$

Model

- Households demand output, given inflation and interest rates:

$$\frac{1}{X_t} = \beta E_t \left[\left(\frac{1}{X_{t+1}} \right) \left(\frac{R_t}{\Pi_{t+1} Z_{t+1}} \right) \right]$$

- Firms set prices, given output demand:

$$\phi + (\Pi_t - 1) = \frac{1}{\phi} X_t + \beta E_t [\Pi_{t+1} - 1]$$

- Central bank sets interest rate, given inflation

$$\beta R_t = \Pi_t^{1/\beta} M_t$$

Model

- The model's control variables are determined by equations:

$$\frac{1}{X_t} = \beta E_t \left[\left(\frac{1}{X_{t+1}} \right) \left(\frac{R_t}{\Pi_{t+1} Z_{t+1}} \right) \right]$$

$$\phi + (\Pi_t - 1) = \frac{1}{\phi} X_t + \beta E_t [\Pi_{t+1} - 1]$$

$$\beta R_t = \Pi_t^{1/\beta} M_t$$

Model

- The model's control variables are determined by equations:

$$\frac{1}{X_t} = \beta E_t \left[\left(\frac{1}{X_{t+1}} \right) \left(\frac{R_t}{\Pi_{t+1} Z_{t+1}} \right) \right]$$

$$\phi + (\Pi_t - 1) = \frac{1}{\phi} X_t + \beta E_t [\Pi_{t+1} - 1]$$

$$\beta R_t = \Pi_t^{1/\beta} M_t$$

- The model is completed by adding equations for the state variables:

$$\ln(Z_{t+1}) = \rho_z \ln(Z_t) + \xi_{t+1}$$

$$\ln(M_{t+1}) = \rho_m \ln(M_t) + e_{t+1}$$

The model in Stata

```
. dsgenl  (1 = {beta}*(F.x/x)^(-1)*(r/(F.p*F.z)))      ///
          ({phi}+(p-1) = 1/{phi}*x + {beta}*(F.p-1))      ///
          ({beta}*r = p^(1/{beta})*m)                      ///
          (ln(F.m) = {rhom}*ln(m))                         ///
          (ln(F.z) = {rhoz}*ln(z))                         ///
          , exostate(z m) observed(p r) unobserved(x)
```

Parameter estimation

```
. dsgenl  (1 = {beta}*(F.x/x)^(-1)*(r/(F.p*F.z)))      ///
>          ({phi}+(p-1) = 1/{phi}*x + {beta}*(F.p-1))    ///
>          ({beta}*r = p^(1/{beta})*m)                   ///
>          (ln(F.m) = {rhom}*ln(m))                   ///
>          (ln(F.z) = {rhoz}*ln(z))                   ///
>          , exostate(z m) observed(p r) unobserved(x)
```

Solving at initial parameter vector ...

Checking identification ...

First-order DSGE model

Sample: 1955q1 - 2015q4
Log likelihood = -753.57131

Number of obs = 244

	OIM					
	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
/structural						
beta	.5146672	.0783493	6.57	0.000	.3611054	.668229
phi	.1659058	.0474002	3.50	0.000	.0730032	.2588083
rhom	.7005483	.0452634	15.48	0.000	.6118335	.789263
rhoz	.9545256	.0186417	51.20	0.000	.9179886	.9910627
sd(e.z)	.650712	.1123897			.4304321	.8709918
sd(e.m)	2.318204	.3047452			1.720914	2.915493

Tests of economic hypotheses

```
. nlcom 1/_b[beta]  
_nl_1: 1/_b[beta]
```

	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]
_nl_1	1.943	.2957884	6.57	0.000	1.363265 2.522735

Policy questions

What is the effect of an unexpected increase in interest rates?

Estimated DSGE model provides an answer to this question. We can subject the model to a shock, then see how that shock feeds through the rest of the system.

Effect on impact: the policy function

```
. estat policy
```


Policy matrix

		Delta-method					
		Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
x	z	2.59502	.9077695	2.86	0.004	.8158242	4.374215
	m	-1.608216	.4049684	-3.97	0.000	-2.401939	-.8144921
p	z	.8462697	.2344472	3.61	0.000	.3867617	1.305778
	m	-.4172522	.0393623	-10.60	0.000	-.4944008	-.3401035
r	z	1.644305	.2357604	6.97	0.000	1.182223	2.106387
	m	.1892777	.0591622	3.20	0.001	.0733219	.3052335

Effect over time: impulse response functions

```
. irf set nkirf.irf, replace  
. irf create model1  
. irf graph irf, impulse(m) response(p x r m) byopts(yrescale) yline(0)
```


Impulse responses from the estimated model




Extracting latent states

- A DSGE model links observed variables to latent state variables through a model
- Once a model's parameters are estimated, latent states can be estimated as well
- `predict state*, state`

Monetary policy state variable



Productivity state variable



Analyzing nonlinear DSGE models

- We can do more than look at impulse responses
- We will switch to a textbook model and explore its features

The stochastic growth model

$$1 = \beta E_t \left[\left(\frac{c_{t+1}}{c_t} \right)^{-1} (1 + r_{t+1} - \delta) \right]$$

$$y_t = z_t k_t^\alpha$$

$$r_t = \alpha z_t k_t^{\alpha-1}$$

$$k_{t+1} = y_t - c_t + (1 - \delta)k_t$$

$$\ln z_{t+1} = \rho \ln z_t + e_{t+1}$$


The stochastic growth model in Stata

```
. dsgenl (1={beta}*(c/F.c)*(1+F.r-{delta}))  
>      (r = {alpha}*y/k)  
>      (y=z*k^{\alpha})  
>      (f.k = y - c + (1-{delta})*k)  
>      (ln(F.z)={rhoz}*ln(z)),  
>      exostate(z) endostate(k) observed(y) unobserved(c r)
```

Data

```
. import fred GDPC1  
. generate dateq = qofd(daten)  
. tsset dateq, quarterly  
. generate lgdp = 100*ln(GDPC1)  
. tsfilter hp y = lgdp
```

Data



Parameter estimation

```
. constraint 1 _b[beta]=0.96
. constraint 2 _b[alpha]=0.36
. constraint 3 _b[delta]=0.025
. dsgenl (1={beta}*(c/F.c)*(1+F.r-{delta}))           ///
>      (r = {alpha}*y/k)                                ///
>      (y=z*k^{\alpha})                                 ///
>      (f.k = y - c + (1-{delta})*k)                 ///
>      (ln(F.z)={rhoz}*ln(z)), constraint(1/3) nocnsreport ///
>      exostate(z) endostate(k) observed(y) unobserved(c r) nolog
Solving at initial parameter vector ...
Checking identification ...

First-order DSGE model
```

Sample: 1947q1 - 2019q1 Number of obs = 289
Log likelihood = -362.93403

y	OIM				
	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]
/structural					
beta	.96	(constrained)			
delta	.025	(constrained)			
alpha	.36	(constrained)			
rhoz	.8391786	.0325307	25.80	0.000	.7754197 .9029375
sd(e.z)	.8470234	.0352336		.7779668	.91608

After parameter estimation

- Long run behavior: steady-state
- Impact effect of shocks: the policy matrix
- How shocks persist over time: the transition matrix
- Exploring the structure: model-implied covariances
- Dynamic effects: impulse responses

Steady-state

- A model consists of a collection of nonlinear dynamic equations
- Under stationarity, in the absence of shocks, the variables in the model converge to a point
- This point is the steady-state and is a vector of numbers that depends on the model parameters

Steady-state

```
. estat steady
```

Location of model steady-state

	Delta-method				
	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]
k	13.94329
z	1
c	2.233508
r	.0666667
y	2.582091

Note: Standard errors reported as missing for constrained steady-state values.

Policy matrix

- A model links current control variables to future control variables, current state variables, and future state variables
- A *solution function* to the model expresses control variables as a function of state variables alone
- The policy matrix is a linear approximation to the solution function
- Example: the model has control variable y_t , state variables (k_t, z_t) , and equation

$$y_t = z_t k_t^\alpha$$

which has (log-)linear approximation

$$\hat{y}_t = \hat{z}_t + \alpha \hat{k}_t$$

Policy matrix

```
. estat policy
```

Policy matrix

		Delta-method					
		Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
c	k	.6371815
	z	.266745	.0244774	10.90	0.000	.2187701	.3147198
r	k	-.64
	z	1
y	k	.36
	z	1

Note: Standard errors reported as missing for constrained policy matrix values.

State transition matrix

- A model describes the evolution of state variables in terms of future control variables, current control variables, current state variables
- A *solution function* to the model expresses future values of state variables as a function of current values of state variables alone
- The state transition matrix is a linear approximation to the solution function
- Example: the log-linear approximation for the transition of z_t is

$$\hat{z}_{t+1} = \rho \hat{z}_t + e_t$$

- Example: the transition equation for capital is

$$k_{t+1} = y_t(k_t, z_t) - c_t(k_t, z_t) + (1 - \delta)k_t$$

where the control variables are expressed as functions of the state variables.

State transition matrix

```
. estat transition
```

Transition matrix of state variables

		Delta-method		z	P> z	[95% Conf. Interval]	
		Coef.	Std. Err.				
F.k	k	.9395996
	z	.1424566	.0039209	36.33	0.000	.1347717	.1501414
F.z	k	0	(omitted)				
	z	.8391786	.0325307	25.80	0.000	.7754197	.9029375

Note: Standard errors reported as missing for constrained transition matrix values.

Model-implied covariances

- A model describes the variances, covariances, and autocovariances of its variables
- `estat covariance` displays these statistics

Model-implied covariances


```
. estat covariance y  
Estimated covariances of model variables
```

	Delta-method					
	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
y var(y)	3.872087	.9694708	3.99	0.000	1.971959	5.772215

Impulse responses

```
. irf set stochirf.irf, replace  
. irf create stochastic_model, step(40)  
. irf graph irf, impulse(z) response(y c k z) yline(0) xlabel(0(4)40)
```

Impulse responses



Sensitivity analysis

- Repeatedly solve the model for different parameter sets
- Explore how changes in parameters affect model output, such as impulse responses

Sensitivity analysis: model setup

```
. local model (1 = {beta}*(F.c/c)^(-1)*(1+F.r-{delta}))    ///
>      (y = z*k^{alpha}))                                         ///
>      (r = {alpha}*y/k)                                         ///
>      (f.k = y - c + (1-{delta})*k)                           ///
>      (ln(f.z) = {rho}*ln(z))
```

Sensitivity analysis: parameter setup

```
. local opts observed(y) unobserved(r c) exostate(z) endostate(k)
. matrix param1 =          (0.96, 0.3, 0.025, 0.9)
. matrix colnames param1 = beta alpha delta rho
. matrix param2 =          (0.96, 0.3, 0.025, 0.7)
. matrix colnames param2 = beta alpha delta rho
. irf set sens.irf, replace
(file sens.irf created)
(file sens.irf now active)
```

Sensitivity analysis: solving with parameter set 1

```
. dsgenl `model', `opts' solve noidencheck from(param1)  
Solving at initial parameter vector ...
```

First-order DSGE model

```
Sample: 1955q1 - 2015q4 Number of obs = 244  
Log likelihood = -2112.1857
```

y	OIM				
	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]
/structural					
beta	.96
delta	.025
alpha	.3
rho	.9
sd(e.z)	1	.		.	.

Note: Skipped identification check.

Note: Model solved at specified parameters; maximization options ignored.

```
. irf create model1, step(40)  
(file sens.irf updated)
```

Sensitivity analysis: solving with parameter set 2

```
. dsgenl `model', `opts' solve noidencheck from(param2)  
Solving at initial parameter vector ...
```

First-order DSGE model

```
Sample: 1955q1 - 2015q4 Number of obs = 244  
Log likelihood = -1829.2761
```

y	OIM				
	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]
/structural					
beta	.96
delta	.025
alpha	.3
rho	.7
sd(e.z)	1	.		.	.


Note: Skipped identification check.

Note: Model solved at specified parameters; maximization options ignored.


```
. irf create model2, step(40)  
(file sens.irf updated)
```

Sensitivity analysis: graphing impulse responses

```
. irf ograph (model1 z y irf, lcolor(blue)) (model2 z y irf, lcolor(red))
```



Full set of impulse responses



Conclusion

- `dsgen1` estimates the parameters of nonlinear DSGE models
- View steady-state, policy matrix, transition matrix
- View model-implied covariances
- Create and analyze impulse responses

Thank You!